<commit_before>#
# urbi-tdl.m4: This file is part of build-aux.
# Copyright (C) 2008, <NAME>.
#
# This software is provided "as is" without warranty of any kind,
# either expressed or implied, including but not limited to the
# implied warranties of fitness for a particular purpose.
#
# See the LICENSE file for more information.
# For comments, bug reports and feedback: http://www.urbiforge.com
#

m4_pattern_forbid([^URBI_])

AC_PREREQ([2.60])

# URBI_LTDL
# ---------
# Use libltdl as shipped by libport, as a convenience library.
#
# To complete the use, one needs to add the following lines in the
# top-level Makefile:
#
# libtool: $(LIBTOOL_DEPS)
#	$(SHELL) ./config.status $@
#
# and use LTDL_CPPFLAGS and LTDL_LIBS where appropriate.
#
# You may also use LTDL_DEPS, but beware that specifying
# PROG_DEPENDENCIES stops Automake from generating dependencies from
# LDADD, which is a nuisance.
#
# Note that using "PROG_LDADD = $(LTDL_LIBS)" using the following
# AC_SUBST'ed LTDL_LIBS is *not* a good idea: because this variable is
# opaque to Automake (could actually be -lFOO etc. which are not valid
# dependencies), it cannot add it as is as a dependency.  As a result,
# if libltdl changes, no relinking will be done.  It is better to
# define LTDL_LIBS in the Makefile.am.
AC_DEFUN([URBI_LTDL],
[# If you need to create a library, use libtool.
LT_PREREQ([2.2.6])
# Name the subdirectory that contains libltdl sources
LT_CONFIG_LTDL_DIR([libltdl])
LT_INIT([pic-only shared disable-static dlopen win32-dll])
LTDL_INIT([nonrecursive])
AC_SUBST([LIBTOOL_DEPS])

AC_SUBST([LTDL_CPPFLAGS], ['$(LTDLINCL)'])
AC_SUBST([LTDL_LIBS],     ['$(LIBLTDL)'])
AC_SUBST([LTDL_DEPS],     ['$(LTDLDEPS)'])

URBI_LIBTOOL
])

## Local Variables:
## mode: autoconf
## End:
<commit_msg>Fix autoconf scripts expansion order.<commit_after>#
# urbi-tdl.m4: This file is part of build-aux.
# Copyright (C) 2008, <NAME>.
#
# This software is provided "as is" without warranty of any kind,
# either expressed or implied, including but not limited to the
# implied warranties of fitness for a particular purpose.
#
# See the LICENSE file for more information.
# For comments, bug reports and feedback: http://www.urbiforge.com
#

m4_pattern_forbid([^URBI_])

AC_PREREQ([2.60])

# Help aclocal finding this macro.
AC_DEFUN([URBI_LTDL], [m4_fatal([aie])])

# URBI_LTDL
# ---------
# Use libltdl as shipped by libport, as a convenience library.
#
# To complete the use, one needs to add the following lines in the
# top-level Makefile:
#
# libtool: $(LIBTOOL_DEPS)
#	$(SHELL) ./config.status $@
#
# and use LTDL_CPPFLAGS and LTDL_LIBS where appropriate.
#
# You may also use LTDL_DEPS, but beware that specifying
# PROG_DEPENDENCIES stops Automake from generating dependencies from
# LDADD, which is a nuisance.
#
# Note that using "PROG_LDADD = $(LTDL_LIBS)" using the following
# AC_SUBST'ed LTDL_LIBS is *not* a good idea: because this variable is
# opaque to Automake (could actually be -lFOO etc. which are not valid
# dependencies), it cannot add it as is as a dependency.  As a result,
# if libltdl changes, no relinking will be done.  It is better to
# define LTDL_LIBS in the Makefile.am.
m4_define([URBI_LTDL],
[# If you need to create a library, use libtool.
LT_PREREQ([2.2.6])
# Name of the subdirectory that contains libltdl sources.
LT_CONFIG_LTDL_DIR([libltdl])
LT_INIT([pic-only shared disable-static dlopen win32-dll])
LTDL_INIT([nonrecursive])
AC_SUBST([LIBTOOL_DEPS])

AC_SUBST([LTDL_CPPFLAGS], ['$(LTDLINCL)'])
AC_SUBST([LTDL_LIBS],     ['$(LIBLTDL)'])
AC_SUBST([LTDL_DEPS],     ['$(LTDLDEPS)'])

URBI_LIBTOOL
])

## Local Variables:
## mode: autoconf
## End:

{'fraction_non_alphanumeric': 0.1144183506280721, 'fraction_numerical': 0.008192244675040962, 'mean_word_length': 4.7234375, 'pattern_counts': {'":': 0, '<': 5, '<?xml version=': 0, '>': 5, 'https://': 0, 'lorem ipsum': 0, 'www.': 2, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '3181208', 'n_tokens_mistral': 1313, 'n_tokens_neox': 1267, 'n_words': 442}